> Math 211 : MAPLE Solution of Assignment #7 -- *Your NAME*

Problem 4: The RSA encription method

```
> restart;
```

(a) Program to encode a given message m using pulic key (n,e) :

```
> encode := proc(m, n, e) local M;
    M := Power(m, e) mod n;
    return(M); end;
```

$$encode := \mathbf{proc}(m, n, e)\ \mathbf{local}\ M;\ M := Power(m, e)\ \mathbf{mod}\ n;\ \mathbf{return}\ M\ \mathbf{end\ proc} \quad (1)$$

Notes: 1) Recall that the command "Power(m,e) mod n" is MAPLE's implementation of the power-mod algorithm which we learned in class. You should not use the "irem" command (i.e. write "irem(m^e, n)", for then the computations take much longer (and you will probably get an overflow error for large values of m and e.

2) A shorter way to write the above program is as follows:

```
> encode := (m, n, e) → Power(m, e) mod n;
```

$$encode := (m, n, e) \rightarrow Power(m, e)\ \mathbf{mod}\ n \quad (2)$$

(b) Test the above program on the message "this is top secret".

Recall from class that this message was translated into the following number list:

```
> ml := [20080919, 91900, 20151600, 19050318, 5200000];
```

$$ml := [20080919, 91900, 20151600, 19050318, 5200000] \quad (3)$$

The public key is:

```
> n := 2364389329; e := 123456787;
```

$$n := 2364389329$$

$$e := 123456787 \quad (4)$$

Thus, using this public key, the encoded message is the following list:

```
> msg := [seq(encode(ml[i], n, e), i = 1..5)];
```

$$msg := [1911594886, 263592852, 1703284834, 284325692, 1976174943] \quad (5)$$

Alternate method: Compute the message blocks individually, and put the results in a list:

```
> M1 := encode(ml[1], n, e); M2 := encode(ml[2], n, e); M3 := encode(ml[3], n, e);
  M4 := encode(ml[4], n, e); M5 := encode(ml[5], n, e); msg := [M1, M2, M3, M4, M5];
```

$$M1 := 1911594886$$

$$M2 := 263592852$$

$$M3 := 1703284834$$

$$M4 := 284325692$$

$$M5 := 1976174943$$

$$msg := [1911594886, 263592852, 1703284834, 284325692, 1976174943] \quad (6)$$

Yet another method: Define (and use) a function encodelist which handles lists:

```
> encodelist := (ml, n, e) → [seq(encode(ml[i], n, e), i = 1..nops(ml))];
```

(7)

$$encodelist := (ml, n, e) \rightarrow \left[seq\left(encode\left(ml_i, n, e \right), i = 1 .. nops(ml) \right) \right] \quad (7)$$

```
> msg := encodelist(ml, n, e);
```

$$msg := [1911594886, 263592852, 1703284834, 284325692, 1976174943] \quad (8)$$

Remark: It is always a good idea to test your programs on data for which you know the answer. For example, if you test your program on the key used in class, then you get:

```
> [seq(encode(ml[i], 101284087, 1234567), i = 1 ..5)];
```

$$[18463460, 81091624, 39290746, 47738594, 77028351] \quad (9)$$

Note that these values agree with those given on the overhead in class.

(c) Procedure to find the secret code (using ifactors):

- use ifactors(n) to factor n and to compute phi(n):

```
> crack := proc(n, e) local p, q, k, f, d, a;
    f := ifactors(n); p := f[2, 1, 1]; q := f[2, 2, 1];
    k := (p - 1)·(q - 1); igcdex(e, k, 'd', 'a');
    d := modp(d, k);
    return(d); end;
```

```
crack := proc(n, e)                                                    (10)
    local p, q, k, f, d, a;
    f := ifactors(n);
    p := f[2, 1, 1];
    q := f[2, 2, 1];
    k := (p - 1) * (q - 1);
    igcdex(e, k, 'd', 'a');
    d := modp(d, k);
    return d
end proc
```

Alternate version (using modp(..) in place of igcdex(..)):

```
> crack := proc(n, e) local p, q, k, f, d;
    f := ifactors(n); p := f[2, 1, 1]; q := f[2, 2, 1];
    k := (p - 1)·(q - 1); d := modp(1/e, k);
    return(d); end;
```

```
crack := proc(n, e)                                                    (11)
    local p, q, k, f, d;
    f := ifactors(n);
    p := f[2, 1, 1];
    q := f[2, 2, 1];
    k := (p - 1) * (q - 1);
    d := modp(1 / e, k);
    return d
end proc
```

(d) Find the secret code d for the example in part (b):

```
> d := crack(n, e);
```

$$d := 1514461963 \quad \textbf{(12)}$$

Use the secret code d to decode the encoded message msg:

```
> dmsg := [seq(encode(msg[i], n, d), i = 1 ..5)];
```

$$dmsg := [20080919, 91900, 20151600, 19050318, 5200000] \quad \textbf{(13)}$$

Check that this is the same as the original message ml:

```
> evalb(dmsg = ml);
```

$$true \quad \textbf{(14)}$$

(e) Deciphering the given encoded message M = [M1, ..., M6] using crack(...):

The encoded message M is:

```
> M := [45435724602850, 32563574741091, 46308572459732, 14201357670970];
```

$$M := [45435724602850, 32563574741091, 46308572459732, 14201357670970] \quad \textbf{(15)}$$

The known public key is:

```
> n_A := 50012305249537; e_A := 22331122334567;
```

$$n\_A := 50012305249537$$

$$e\_A := 22331122334567 \quad \textbf{(16)}$$

Step 1: Find the secret key d (using the program "crack"):

```
> d := crack(n_A, e_A);
```

$$d := 14759379960263 \quad \textbf{(17)}$$

Step 2: Decode the given encoded message:

```
> msg2 := [seq(encode(M[i], n_A, d), i = 1 ..4)];
```

$$msg2 := [12052000211900, 7150020150001, 160118202500, 20151409070820] \quad \textbf{(18)}$$

Step 3: Interpret (by hand) the above message:

Break each number into a sequence of numbers consisting of 2 digits (work from right to left):

12052000211900 =12|05|20|00|21|19|00 = let*us*

where * represents a blank (space). Similarly, the other numbers are converted, padding with blanks, if necessary. Thus we get:

12052000211900 = 12|05|20|00|21|19|00 = let*us*
7150020150001 = 7|15|00|20|15|00|01 = go*to*a
160118202500 = 00|16|01|18|20|25|00 = *party*
20151409070820 = 20|15|14|09|07|08|20 = tonight

and so the message is:

"let us go to a party tonight".

Note (for MAPLE experts): We can also use MAPLE to translate our decoded message into words by using the following decoding procedure which converts a list of numbers (viewed as character blocks of length b) into a string:

```
> converttostring := proc(lis, b)
  local blocks, i, j, l, m, r, alphabet, str, bl;
  l := nops(lis); alphabet := `abcdefghijklmnopqrstuvwxyz`; str := ``;
  for i to l do; m := lis[i]; bl := ``;
    for j from 1 to b do; r := modp(m, 100); m := (m-r)/100;
      if r < 1 or r > 26 then bl := cat(` `, bl)
                         else bl :=
```

```
        cat(substring(alphabet, r..r), bl); fi; od;
      str := cat(str, bl); od;
    return(str); end:
```

However, before being able to use this program, we have to figure out what block-length was used. From the above output msg2 in Step 2 we see that all the numbers have 12 to 14 decimal digits, which means that b = 7. Applying the above program to msg2 we obtain:

```
> converttostring(msg2, 7);
```

let us go to a party tonight **(19)**

This, therefore, gives the same message as we had worked out by hand.

Note that if we had used the wrong block length, then we would have either lost letters or had inserted extra blanks:

```
> converttostring(msg2, 5); converttostring(msg2, 9);
```

t us to aarty night

let us go to a party tonight **(20)**

Comment (for MAPLE experts): Similarly, we could use MAPLE to translate a given text into a sequence of number blocks by using the following conversion program "converttoblocks" (which uses the following program "converttonumberlist"):

(i) converttonumberlist: converts a string "str" to a list of numbers:

```
> converttonumberlist := proc(str) local lis, i; lis := [ ];
    for i to length(str) do;
      lis := [op(lis), searchtext(substring(str, i..i),
          abcdefghijklmnopqrstuvwxyz) ]; od;
    return(lis); end:
```

For example, the text "This is top secret"which was used in class is converted as follows:

```
> converttonumberlist( `This is top secret`);
```

[20, 8, 9, 19, 0, 9, 19, 0, 20, 15, 16, 0, 19, 5, 3, 18, 5, 20] **(21)**

(ii) converttoblocks: converts a string into list of numbers corresponding to blocks of length b:

```
> converttoblocks := proc(str, b)
    local lis, blocks, i, j, l, r;
    lis := converttonumberlist(str);
    l := nops(lis); r := irem(l, b);
    if r ≠ 0 then for i to b-r do;
      lis := [op(lis), 0]; od; fi;
    l := nops(lis) / b; blocks := [ ];
    for i from 0 to l - 1 do;
      r := lis[b * i + 1];
      for j from 2 to b do; r := r * 100 + lis[b * i + j]; od;
        blocks := [op(blocks), r]; od;
    return(blocks); end:
```

Thus, the message used in class is translated as follows:

```
> m := converttoblocks( `this is top secret`, 4);
```

m := [20080919, 91900, 20151600, 19050318, 5200000] **(22)**

```
>
```